
SYSTEMS OF DISTRIBUTIONS 

The Johnson Distributions 
If a theoretical distribution is to estimate some empirical distribution, it must be able to, to 
some degree, to recreate location, scale and shape, and it must be possible to estimate the 
parameters of the theoretical distribution from the data. The Johnson’s system (Johnson and 
Kotz 1972) consist of three non-linear transformations that cover the entire space of shapes, 
when shape is defined as a point in the skewness X kurtosis space.  
Let Z be a standard Normal variate. Let X be the non-linear transformations given by: 

inverse 
ZX sinh − γ = ξ + λ ⋅  δ 

 

Johnson called these three transformations System bounded (SB), System lognormal (SL) and 
system unbounded (SU), respectively. SL does not have a λ parameter since it could not be 
separated from γ. The linear transformation Z X= γ + δ ⋅ (Normal distribution) is a limiting 
case of all three. The idea of this system is to take benefit of the normal theory on the Z scale. 
SU has not been relevant in forestry. 
(Lambert 1970) and partly revisited by(Rennolls and Wang 2005) introduced the following 
parameterization of SB, which is statistically better in (µ,σ) and numerically symmetric in 
lower and upper border (τ,θ). This parameterization is used here. 

Estimating SB is elaborate. However, once (τ,θ) is found, (µ,σ) is readily available as in 
normal distribution. (Mønness 1982) proposed starting values for the iterative maximum 
likelihood procedure. (Johnson 1949) introduced estimating (τ,θ,µ,σ) by fractiles. However, 
the fractiles have to obey certain restrictions (Mønness 1982), revisited in (Siekierski 1992), 
so that not any set of fractiles yields a SB distribution. The full maximum likelihood method 
is complicated (appendix A1). (Hahn and Shapiro 1967) indicate that the quality of the fit is 
relatively unaffected by the choice of lower and upper bound as long as they are reasonable. 

However, the xlog
1 x
 
 − 

 transformation on x 0,1∈  (basis of SB) is symmetric and rather 

linear in the interior, say x 0.2,0.8∈ . If upper and lower bound is chosen symmetric and 
somehow away from the data the transformation is close to linear and yields no change of 
shape. Thus quality estimates of (τ,θ) is crucial for SB. The log-Likelihood surface on (τ,θ) is 
rather flat around the maximizing value, this can imply convergence problems . Fixing the 

lower border τ improves the numeric convergence. A SAS program for ML estimation is 
given in appendix A4. 
 

The Power-Normal distribution 
The Box-Cox transformation is applicable for positive data, X≥0.  

Z

=

( )log X

The transformation is continuous in λ so estimation involves only the first functional form. 
The transformation can always be done, but Z can only be N(0,1) in the log() case, and the 
obvious case λ=1. However, X is near normal (Hernandez and Johnson 1980), actually it is a 
truncated normal (Freeman and Modarres 2006).  
The transformed value is supposed to be normal (λ≠0): 

( ) ( )
x 1

However, this is not a true cumulative distribution since the Φ() does not cover (0,1) since: 

λ
−∞−∈

λ
−λ 1,1X  when λ<0 and 

(When λ=0, the entire real line is covered). 
The distribution is  

When λ>0, K is typically close to 1 (often on real data µ>>σ). When λ<0, but close to zero, K 
is still close to 1. If λ<0 and moves away from zero, K becomes small and the truncation does 

influence. However, large absolute values of λ are of little practical value since then the 
transformation becomes close to one vertical and one horizontal line intersection in (1,0). 
The PN distribution has only one functional form, thus engaging only one maximum 
likelihood estimation procedure. If the estimated λ is close to zero one can change to the log-
normal. Typically one ignores including K in the estimation step. A maximum likelihood 
procedure ignoring K is described in (Madansky 1988). The estimation is well-established 
and with good numerical properties. A SAS program is given in appendix A5. 
When prediction a distribution, one has to include K in order to have a proper distribution. 
The shapes of the PN is explored in the main article (Mønness 2011). 
 
Both SB and PN has the lognormal distribution as a limiting case 

Prediction a distribution from a set of stands.. 
A typical forestry task is to first estimate parameters on each stand. Second, one uses the 
estimated distribution as a dependent observation in a regression, with stand characteristics as 
independent variables, in order to make predictions on future development and on other 
stands. Independent characteristics are typically site index, age, density, location/climate, 
basal area mean diameter etc. (Basal area mean diameter is typically an independent 
characteristics since it can be judged on site without knowing all the diameters, but it is also a 

property of the diameter distribution; ( )2E D . ) 

 The estimated parameters typically perform poorly as dependent variables since they may not 
act well in a linear model with stand characteristics. Also, different sets of parameters can 
yield nearly equal shapes. An alternative is to, from the estimated distributions, to calculate a 
certain set of fractiles that are used as dependent observations in the regression. After this 
second estimation, one must invert the calculations to get parameters.  
 
With SB this was done by (Mønness 1982). A problem, mentioned earlier, is that not every 
set of 4 fractiles (minimum, median and two symmetrically chosen fractiles) can yield a SB 
distribution (appendix A2). With PN, any set of three fractiles (median and two symmetrically 
chosen fractiles) yields PN parameter estimates (appendix A3). 

 A1. MAXIMUM LIKELIHOOD OF SB. 

Here the complete ML equations are presented. The parameterization is from (Lambert 1970). 
The likelihood of a set of n independent observations from a SB is 

( ) ( )
( ) ( ) ( )

2n
i

1 n n
i i2

i i
i

1 xL x ,..., x ; , , , exp log
2 x2 x x

  θ − τ  − − τ µ σ τ θ = ⋅ −µ   σ θ −   πσ − τ θ −  
∑

∏
 

Partial derivatives of σ  and µ , given θ  and τ  are the usual mean and standard deviation of a 
normal distributed data set: 

The log likelihood to be maximized by Newton’s method is then 

These formulas involve 18 different sums of the observations (Appendix A4). Initial values 
are discussed in (Mønness 1982). If τ is forced to 0 or another fixed value, this value can be 
substituted everywhere and then only the θ derivatives are used.  

A2. SB FROM FRACTILES. 

Johnson (1949) indicated the estimating procedure that also can be used as a means to go 
from predicted fractiles to a predicted distribution. Suppose four fractiles is given or is 
predicted from a regression.  Then ( )(0) (p) (0.5) (1 p)X ,X ,X ,X −  can give a closed solution of θ as 
given below.  
Let p (p) (0)f X X= − , (the lower p-percentile minus the lower bound), 1 p (1 p) (0)f X X− −= −  and 

0.5 (0.5) (0)f X X= −  (the median minus the lower bound). Then (0)Xτ =  and  

0.5 p 0.5 1 p p 1 p
0.5 2

0.5 p 1 p

f f f f 2f f
f

f f f
− −

−

+ −
θ = τ + ⋅

−
. 

The denominator has to be positive. Thus not all possible set of percentiles can give raise to a 
SB. µ and σ are readily available. 

Let x f
f p

=
−

0 5

1

.  and y f
f p

= 0 5. . Then the θ-τ expression can be reformulated as a function of 

the kind x y
x y
+ −
⋅ −

2
1

 with 0 x 1 y< < < . This function has singularities along xy=1 and can give 

unstable solutions. 

A3. PN FROM FRACTILES. 

Suppose the (diameter or height) fractiles ( ) ( ) ( )p 0.50 1 pX ,X ,X −  are given (from prediction 

regressions or from actual real data). In order to find the parameter set λ µ σ, ,  the following 
equations have to be solved: 

( ) ( )

1 1

X 1 z

X 1 z 1

 where the z's are fractiles from the standard normal distribution. 
λ  can be solved from 

, which has a zero solution in addition to the solution of interest. 

The first equation has no explicit solution but can be solved numerically.  

The minimum value is explicitly determined by  

If 
0 0λ < then solution 0λ < λ  

0 0λ = then solution 0λ =  

0 0λ > then solution 0λ > λ  
Thus starting values of the numeric solution are determined. Newton’s method (Henrici 1964) 
is 

( )
( )

n
n 1 n

n

F
F+

λ
λ = λ −

′ λ
 until convergences.  

The use of the median is critical to solve the system.  

A4. SAS PROGRAM ML FOR SB 

The SAS program (SAS 2008) for maximum likelihood estimation. In the progam, sigma is 
named beta  
/*----------------------------------------------------------------------- 
                            Beregning av parametre i Johnson Sb. 
                              for bivariat fordeling diametre og 
                              høyder. 
                            estimeringsmetode ML med nedre 
                                                   (tau)=xmin-faktor0 
                              desember 2009. 
                              fil= hoydiam.dhparJSbf_09.sas 
                             Beregner parametre og legger ut enkeltrær 
----------------------------------------------------------------------*/ 
data hoydiam.dhparJSbf  
(keep=flate dtau dteta dbeta dmy dloglikl ditr dispes 
            htau hteta hbeta hmy hloglikl hitr hispes  
                             n     ro) 
     hoydiam.JSbf_dh (keep=flate lfd lfh)               ; 
 
set hoydiam.dh ;by flate; 
 
array dd{550} 3 _temporary_;                   /*  enkelt-dataverdier */ 
array hh{550} 3 _temporary_;                   /*  enkelt-dataverdier */ 
array f{18} _temporary_;                       /*  f(hver obs)  */ 
array s{18} _temporary_;                       /*   "kvadratsummer" */ 
retain n dmin dmax hmin hmax   dd hh ; 
/*------------------------------------innlesing av alle trær--------------*/ 
 
if first.flate 
then           do;                              /*initier tellere */ 
                 n=1; 
                 dmin=diam ; dmax=diam ; dd{1}=diam ; 
                 hmin=hoyde; hmax=hoyde; hh{1}=hoyde; 
               end; 
else           do;                              /* les inn en flate */ 
                 n=n+1; 
                 dd{n}=diam; hh{n}=hoyde; 
                        if diam <dmin then dmin=diam ; 
                  else  if diam >dmax then dmax=diam ; 
                        if hoyde<hmin then hmin=hoyde; 
                  else  if hoyde>hmax then hmax=hoyde; 
               end; 
/*--------------------------------------------------beregn for en flate */ 
if last.flate then do; 
/*----------------------------------------diametre--------------------*/ 
 
   ispes=0; itr=0; 
                                /* startverdier iterasjon */ 
   faktor=(dmax - dmin)/sqrt(n); 
/*   tau=max(0,dmin-faktor)*/ 
   tau= 0; 
   teta=dmax+2*faktor; 
 
   do while (ispes=0 and itr< 50                  ); 
 
   do j=1 to 18; s{j}=0; end;            /* nullstiller summer */ 
                 do i=1 to n; 
                            f{ 1}=1/(dd{i}-tau); 
                            f{ 2}=1/(teta - dd{i} ); 
 
                            f{ 3}=f{ 1}*f{ 1}; 

                            f{ 4}=f{ 2}*f{ 2}; 
                            f{ 5}=f{ 1}*f{ 2}; 
                            f{ 6}=  log(dd{i}-tau); 
                            f{ 7}=  log(teta-dd{i}); 
                            f{ 8}=f{ 6}*f{ 7}; 
                            f{ 9}=f{ 6}*f{ 6}; 
                            f{10}=f{ 7}*f{ 7}; 
                            f{11}=f{ 6}*f{1}; 
                            f{12}=f{ 7}*f{1}; 
                            f{13}=f{ 6}*f{2}; 
                            f{14}=f{ 7}*f{2}; 
                            f{15}=f{ 6}*f{3}; 
                            f{16}=f{ 7}*f{3}; 
                            f{17}=f{ 7}*f{4}; 
                            f{18}=f{ 6}*f{4}; 
 
                            do j=1 to 18; s{j}=s{j} + f{j}; end; 
                            end; 
 
 
    beta=( s{9} + s{10} - 2*s{8} -( s{6}*s{6} + s{7}*s{7} -2*s{6}*s{7}) 
         /n )/n; 
    my=    (s{6} - s{7})/n; 
 
                if itr=0 then do; 
/*                                 tau0= max(0,dmin-faktor)*/ 
                                 tau0= 0; 
                                 teta0=dmax+2*faktor; 
                                 my0=my; 
                                 beta0=beta; 
                                 s60=s{6}; 
                                 s70=s{7}; 
                                 end; 
 
                                                   /* log(2*pi*e) */ 
    bDtet  =2*( s{14} - s{13} + (s{6} - s{7})*s{2}/n  )/n; 
    bDtettet =2* ( s{4} - s{17} + s{18} -( s{2}*s{2} - s{7}*s{4} + 
                                          s{4}*s{6} )/n  )/n; 
 
 
    LDtet= -s{2} + n/(teta-tau) - n*bDtet/(2*beta); 
 
    LDtettet= s{4} - n/((teta-tau)*(teta-tau)) 
                   +(n/2)*( bDtet*bDtet/(beta*beta) - bDtettet/beta); 
 
            if LDtettet NE 0 then do; 
               del_tet=LDtet/ LDtettet; 
 
                       if      teta - del_tet > dmax +0.5 
                           THEN do; 
                                endring=sqrt((del_tet**2) 
                                            /( teta**2)); 
                                if endring > 0.001 then do; 
                                                    itr=itr+1; 
                                                    teta=teta-del_tet; 
                                                    end; 
                                             ELSE   ispes=1; 
                                end; 
                           ELSE ispes=2; 
                       end; 
                       else ispes=3; 
     end;  /* end while*/ 
    if (ispes NE 1) OR (teta>2*teta0) then do; 
                                        /* bruker startverdier */ 
                                        tau= tau0; 
                                        teta=teta0; 
                                        beta=beta0; 
                                        my=my0; 
                                        s{6}=s60; 
                                        s{7}=s70; 
                                        ispes=10+ispes; 
                                        end; 
 
    loglikli= n*log(teta-tau) - (n/2)*log(beta) - s{6} -s{7} - 
                               (n/2)*2.837877066409; 
    dbeta=  beta; 
    dmy=    my; 

dtau=tau; dteta=teta; 
dn=n;                      dloglikl=loglikli; dispes=ispes; ditr=itr; 
/*------------------------------------------høyder--------------------*/ 
   ispes=0; itr=0; 
                                /* startverdier iterasjon */ 
   faktor=(hmax - hmin)/sqrt(n); 
   tau= max(0,hmin-faktor); 
   teta=hmax+faktor; 
 
   do while (ispes=0 and itr< 50                  ); 
 
   do j=1 to 18; s{j}=0; end;            /* nullstill summer */ 
                 do i=1 to n; 
                            f{ 1}=1/(hh{i}-tau); 
                            f{ 2}=1/(teta - hh{i} ); 
 
                            f{ 3}=f{ 1}*f{ 1}; 
                            f{ 4}=f{ 2}*f{ 2}; 
                            f{ 5}=f{ 1}*f{ 2}; 
                            f{ 6}=  log(hh{i}-tau); 
                            f{ 7}=  log(teta-hh{i}); 
                            f{ 8}=f{ 6}*f{ 7}; 
                            f{ 9}=f{ 6}*f{ 6}; 
                            f{10}=f{ 7}*f{ 7}; 
                            f{11}=f{ 6}*f{1}; 
                            f{12}=f{ 7}*f{1}; 
                            f{13}=f{ 6}*f{2}; 
                            f{14}=f{ 7}*f{2}; 
                            f{15}=f{ 6}*f{3}; 
                            f{16}=f{ 7}*f{3}; 
                            f{17}=f{ 7}*f{4}; 
                            f{18}=f{ 6}*f{4}; 
 
                            do j=1 to 18; s{j}=s{j} + f{j}; end; 
                            end; 
 
 
    beta=( s{9} + s{10} - 2*s{8} -( s{6}*s{6} + s{7}*s{7} -2*s{6}*s{7}) 
         /n )/n; 
    my=    (s{6} - s{7})/n; 
 
                if itr=0 then do; 
                                 tau0= max(0,hmin-faktor); 
                                 teta0=hmax+faktor; 
                                 my0=my; 
                                 beta0=beta; 
                                 s60=s{6}; 
                                 s70=s{7}; 
                                 end; 
 
                                                   /* log(2*pi*e) */ 
    bDtet  =2*( s{14} - s{13} + (s{6} - s{7})*s{2}/n  )/n; 
    bDtettet =2* ( s{4} - s{17} + s{18} -( s{2}*s{2} - s{7}*s{4} + 
                                          s{4}*s{6} )/n  )/n; 
 
 
    LDtet= -s{2} + n/(teta-tau) - n*bDtet/(2*beta); 
 
    LDtettet= s{4} - n/((teta-tau)*(teta-tau)) 
                   +(n/2)*( bDtet*bDtet/(beta*beta) - bDtettet/beta); 
 
            if LDtettet NE 0 then do; 
               del_tet=LDtet/ LDtettet; 
 
                       if      teta - del_tet > hmax +0.5 
                           THEN do; 
                                endring=sqrt((del_tet**2) 
                                            /( teta**2)); 
                                if endring > 0.001 then do; 
                                                    itr=itr+1; 
                                                    teta=teta-del_tet; 
                                                    end; 
                                             ELSE   ispes=1; 
                                end; 
                           ELSE ispes=2; 
                       end; 
                       else ispes=3; 

     end;  /* end while*/ 
    if (ispes NE 1) OR (teta>2*teta0) then do; 
                                        /* bruker startverdier */ 
                                        tau=tau0; 
                                        teta=teta0; 
                                        beta=beta0; 
                                        my=my0; 
                                        s{6}=s60; 
                                        s{7}=s70; 
                                        ispes=10+ispes; 
                                        end; 
 
    loglikli= n*log(teta-tau) - (n/2)*log(beta) - s{6} -s{7} - 
                               (n/2)*2.837877066409; 
    hbeta=  beta; 
    hmy=    my; 
htau=tau; hteta=teta; 
hn=n;                      hloglikl=loglikli; hispes=ispes; hitr=itr; 
/*---------------------------bivariate parametre og utlegg av enkelttrær--*/ 
        ro=0; 
        do i=1 to n; 
               lfd=(log((dd{i}-dtau)/(dteta-dd{i})) -dmy)/sqrt(dbeta); 
               lfh=(log((hh{i}-htau)/(hteta-hh{i}))-hmy)/sqrt(hbeta);; 
                ro=ro+lfd*lfh; 
                output hoydiam.JSbf_dh; 
        end; 
        ro=ro/n; 
/*        teta=exp((ro*dgamma-hgamma)/hdelta); 
        fi=ro*ddelta/hdelta;*/ 
 
/*-----------------------------------------utlegging parametre------------*/ 
output hoydiam.dhparJSbf; 
put 'flate= ' flate ' ro= ' ro; 
end; /*last.flate*/ 
 
 run; 

A5. SAS PROGRAM ML FOR PN 

The SAS program (SAS 2008) for maximum likelihood estimation  
/*----------------------------------------------------------------------- 
                                      Beregning av parametre i BoxCox 
                                      for bivariat fordeling diametre og 
                                      høyder. 
                                      estimeringsmetode s178 Albert Madansky 
                                     juni 1993. 
                                      fil= hoydiam\dhbox1.sas 
-------------------------------------------------------------------------*/ 
data hoydiam.dhbox1 (keep= flate dbmy dbsigma dlanda hbmy hbsigma hlanda ro) 
     hoydiam.box_dh (keep=flate box_d box_h)               ; 
 
set hoydiam.dh (obs=300);by flate; 
 
array x_d{550} 3 _temporary_;                   /*  enkelt-dataverdier */ 
array x_h{550} 3 _temporary_;                   /*  enkelt-dataverdier */ 
array yl_d{550} _temporary_; 
array yl_h{550} _temporary_; 
array dy_d{550} _temporary_; 
array dy_h{550} _temporary_; 
retain n x_d x_h yl_d yl_h dy_d dy_h yx_d yx_h    ; 
/*------------------------------------innlesing av alle trær--------------*/ 
 
if first.flate 
then           do;                              /*initier tellere */ 
                 n=1; 
                 x_d{1}=diam ; 
                 x_h{1}=hoyde; 
                yx_d=log(diam); yx_h=log(hoyde); 
               end; 
else           do;                              /* les inn en flate */ 
                 n=n+1; 
                 x_d{n}=diam; x_h{n}=hoyde; 
                yx_d=yx_d+log(diam); yx_h=yx_h+log(hoyde); 
               end; 
/*--------------------------------------------------beregn for en flate */ 
if last.flate then do; 
/*--------------------------------------------diametre--------------------*/ 
xlanda_d=0.25; 
wz=1; 
do while (wz>0.000005); 
        do i=1 to n; 
                yl_d{i}=(x_d{i}**xlanda_d -1)/xlanda_d; 
                end; 
        ym_d=0; ys_d=0; 
        do i=1 to n; 
                ym_d=ym_d+yl_d{i}/n;                 /*y_landa mean */ 
                ys_d=ys_d+yl_d{i}*yl_d{i}; 
                end; 
                ys_d=(ys_d-n*ym_d*ym_d)/n;               /* var y_landa  */ 
        z_d=0; xlx_d=0; yxlx_d=0; 
        do i=1 to n; 
                xlx_d=xlx_d+log(x_d{i})*x_d{i}**xlanda_d; 
                yxlx_d=yxlx_d + yl_d{i}*log(x_d{i})*x_d{i}**xlanda_d ; 
                z_d=z_d+(yl_d{i}-ym_d)*((1+xlanda_d*yl_d{i})* 
                        log(1+xlanda_d*yl_d{i}) - xlanda_d*yl_d{i}); 
                end; 
                z_d=yx_d-z_d/(xlanda_d*xlanda_d*ys_d); 
        do i=1 to n; 
                dy_d{i}=(-yl_d{i}+x_d{i}**xlanda_d * log(x_d{i}))/xlanda_d; 
                end; 
        dm_d=0; 
        do i =1 to n; 
                dm_d=dm_d+dy_d{i}/n; 
                end; 
        dsig_d=-2*ys_d/xlanda_d +2*(yxlx_d-xlx_d*ym_d)/(xlanda_d*n); 

        g_d= -(z_d-yx_d)*(2*xlanda_d*ys_d + 
                xlanda_d*xlanda_d*dsig_d)/(xlanda_d*xlanda_d*ys_d); 
        s2_d=0; s1_d=0; 
        do i =1 to n; 
                wy_d=1+xlanda_d*yl_d{i}; 
                s1_d=s1_d + 
                     (yl_d{i}-ym_d)*log(wy_d)*(yl_d{i}+xlanda_d*dy_d{i}); 
                s2_d=s2_d +(dy_d{i}-dm_d)*(wy_d*log(wy_d)-xlanda_d*yl_d{i}); 
                end; 
        g_d=g_d-(s1_d+s2_d)/(xlanda_d*xlanda_d*ys_d); 
        put flate xlanda_d z_d g_d; 
        wz=abs(z_d); 
        if wz>0.000005 then xlanda_d=xlanda_d-z_d/g_d; 
 
end; 
 
/*--------------------------------------------høyder--------------------*/ 
xlanda_h=0.25; 
wz=1; 
do while (wz>0.000005); 
        do i=1 to n; 
                yl_h{i}=(x_h{i}**xlanda_h -1)/xlanda_h; 
                end; 
        ym_h=0; ys_h=0; 
        do i=1 to n; 
                ym_h=ym_h+yl_h{i}/n;                 /*y_landa mean */ 
                ys_h=ys_h+yl_h{i}*yl_h{i}; 
                end; 
                ys_h=(ys_h-n*ym_h*ym_h)/n;               /* var y_landa  */ 
        z_h=0; xlx_h=0; yxlx_h=0; 
        do i=1 to n; 
                xlx_h=xlx_h+log(x_h{i})*x_h{i}**xlanda_h; 
                yxlx_h=yxlx_h + yl_h{i}*log(x_h{i})*x_h{i}**xlanda_h ; 
                z_h=z_h+(yl_h{i}-ym_h)*((1+xlanda_h*yl_h{i})* 
                        log(1+xlanda_h*yl_h{i}) - xlanda_h*yl_h{i}); 
                end; 
                z_h=yx_h-z_h/(xlanda_h*xlanda_h*ys_h); 
        do i=1 to n; 
                dy_h{i}=(-yl_h{i}+x_h{i}**xlanda_h * log(x_h{i}))/xlanda_h; 
                end; 
        dm_h=0; 
        do i =1 to n; 
                dm_h=dm_h+dy_h{i}/n; 
                end; 
        dsig_h=-2*ys_h/xlanda_h +2*(yxlx_h-xlx_h*ym_h)/(xlanda_h*n); 
        g_h= -(z_h-yx_h)*(2*xlanda_h*ys_h + 
                xlanda_h*xlanda_h*dsig_h)/(xlanda_h*xlanda_h*ys_h); 
        s2_h=0; s1_h=0; 
        do i =1 to n; 
                wy_h=1+xlanda_h*yl_h{i}; 
                s1_h=s1_h + 
                     (yl_h{i}-ym_h)*log(wy_h)*(yl_h{i}+xlanda_h*dy_h{i}); 
                s2_h=s2_h +(dy_h{i}-dm_h)*(wy_h*log(wy_h)-xlanda_h*yl_h{i}); 
                end; 
        g_h=g_h-(s1_h+s2_h)/(xlanda_h*xlanda_h*ys_h); 
        put flate xlanda_h z_h g_h; 
        wz=abs(z_h); 
        if wz>0.000005 then xlanda_h=xlanda_h-z_h/g_h; 
end; 
/*---------------------------bivariate parametre og utlegg av enkelttrær--*/ 
 
        ro=0; 
                dlanda=xlanda_d; hlanda=xlanda_h; 
                dbsigma=sqrt(ys_d); dbmy=ym_d; 
                hbsigma=sqrt(ys_h); hbmy=ym_h; 
        do i=1 to n; 
                box_d=(yl_d{i}-dbmy)/dbsigma; 
                box_h=(yl_h{i}-hbmy)/hbsigma; 
               ro=ro+box_d*box_h; 
                output hoydiam.box_dh; 
        end; 
        ro=ro/n; 
 
/*-----------------------------------------utlegging parametre------------*/ 
 
output hoydiam.dhbox1; 
put 'flate= ' flate ' ro>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>= ' ro; 

put dlanda hlanda; 
end; /*last.flate*/ 
 
 run; 

